Pensieve header: Kauffman @ A cubic root of -1.

In[∘]:=
```
<< KnotTheory`
```

Loading KnotTheory` version of February 2, 2020, 10:53:45.2097.
Read more at http://katlas.org/wiki/KnotTheory.

In[∘]:=
```
KB[pd_PD] := Module[{p, t1, t2, t3, t4, B, d, KB, todo, front, x, v},
   SetAttributes[p, Orderless];
   KB = 1;
   todo = List @@ pd;
   front = {};
   v[x_X] := Length[front ⋂ (List @@ x)];
   While[Length[todo] > 0,
    x = RandomChoice[MaximalBy[todo, v]];
    todo = DeleteCases[todo, x];
    t1 = KB (x /. X[i_, j_, k_, l_] → A * p[i, j] * p[k, l] + B * p[i, l] * p[j, k]);
    t2 = Expand[t1];
    t3 = t2 //. {p[i_, j_] * p[j_, k_] → p[i, k]};
    t4 = t3 /. {p[i_, i_] → d, p[i_, j_]^2 → d};
    KB = Expand[t4 /. {B → 1/A, d → -A^2 - 1/A^2}];
    front = Complement[front ⋃ (List @@ x), front ⋂ (List @@ x)];
   ];
   KB
  ];
KB[K_] := KB[PD[K]]
```

In[∘]:=
```
Short[tabK = KB /@ AllKnots[{3, 8}]]
```

Out[∘]//Short=

$$\left\{\frac{1}{A^7}+\frac{1}{A^3}+A-A^9,\ -\frac{1}{A^{10}}-A^{10},\ \frac{1}{A^9}+\frac{1}{A^5}+\frac{1}{A}-A^{15},\ \ll 30\gg,\ -1+\frac{1}{A^{12}}-\frac{1}{A^8}-\frac{1}{A^4}-A^4+A^{16},\ -\frac{2}{A^{10}}-\frac{1}{A^2}+A^6+A^{14}-A^{18}\right\}$$

In[∘]:=
```
Union@Simplify[tabK /. A → e^(π i/3)]
```

Out[∘]= {1}

In[∘]:=
```
Short[tabL = KB /@ AllLinks[{2, 10}]]
```

Out[∘]//Short=

$$\left\{\frac{1}{A^6}+\frac{1}{A^2}+A^2+A^6,\ 1+\frac{1}{A^{12}}+A^4+A^8,\ \ll 414\gg,\ -5-\frac{1}{A^{20}}-\frac{4}{A^{12}}-\frac{4}{A^8}-\frac{4}{A^4}-5A^4-6A^8-A^{12}-A^{16}-A^{20}\right\}$$

In[∘]:=
```
Union@Simplify[tabL /. A → e^(π i/3)]
```

Out[∘]= {1}

In[∘]:= $\left(e^{\pi i/3}\right)^3$

Out[∘]= -1

In[∘]:= $\left(e^{\pi i/3}\right)^{-4}$

Out[∘]= $e^{\frac{2 i \pi}{3}}$

In[ ]:= $\left(e^{\pi i/3}\right)^{-4} / \left(e^{\pi i/3}\right)^{4}$

Out[ ]= $e^{-\frac{2 i \pi}{3}}$

In[ ]:= `N[(-A^2 - A^-2) /. A -> E^(π I/3)]`

Out[ ]= 1. + 0. i